Modeling the World Population as Logistic Growth

Helmut Knaust, Department of Mathematical Sciences, UTEP, El Paso TX 79968
hknaust@utep.edu
12/30/2009; last edit 11/2/2011

The World Population

Here is a table for the world population from 1950-2005 and its graph:

In[1]:=

```
popy = Table[k, {k, 1950, 2005, 5}];
popd =
  {2.519, 2.756, 2.982, 3.335, 3.692, 4.068, 4.435, 4.831, 5.263, 5.674, 6.070, 6.454};
pop = Transpose[{popy, popd}];
TableForm[pop, TableHeadings → {None, {"Year", "Population(10^9)"}},
 TableAlignments → Right]
p1 = ListPlot[pop, PlotStyle → {Red, AbsolutePointSize[7]}, BaseStyle → {Bold, 14},
  PlotRange → {0, 7}, AxesOrigin → {1950, 0}, PlotLabel → "World Population 1950-2005",
  AxesLabel → {"Year", "Population (10^9)"}, GridLines → Automatic, ImageSize → 500]
```

Out[4]//TableForm=

Year	Population(10^9)
1950	2.519
1955	2.756
1960	2.982
1965	3.335
1970	3.692
1975	4.068
1980	4.435
1985	4.831
1990	5.263
1995	5.674
2000	6.07
2005	6.454

Out[5]=



The Logistic Growth Equation

Next we compute the solution to the logistic growth equation with parameters r, K and initial population P.

In[6]:=
```
Clear[p];
eqn = p'[t] == r p[t] (1 - p[t] / K)
```

Out[7]=
$$p'(t) = r\, p(t) \left(1 - \frac{p(t)}{K}\right)$$

A typical solution to a logistic growth equation:

In[8]:=
```
Plot[10 Exp[.2 t] / (400 + Exp[.2 t]), {t, 10, 55}]
```

Out[8]=


In[9]:=
```
sol = DSolve[{eqn, p[0] == P}, p[t], t]
```

Solve::ifun : Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information. ≫

Out[9]=
$$\left\{\left\{p(t) \to \frac{K P e^{r t}}{K + P e^{r t} - P}\right\}\right\}$$

In[10]:=
```
p[t_] = p[t] /. sol[[1]]
```

Out[10]=
$$\frac{K P e^{r t}}{K + P e^{r t} - P}$$

Using the Method of Least Squares to Estimate the Parameters r and K

We use Newton' s Method to minimize the least squares error:

In[11]:=
```
<< Optimization`UnconstrainedProblems`
```

In[12]:=
```
ls[r_, K_, P_] = Total[(Table[p[t], {t, 0., 55, 5}] - popd)^2]
```

Out[12]=

$$\left(\frac{K P e^{5.r}}{K + P e^{5.r} - P} - 2.756\right)^2 + \left(\frac{K P e^{10.r}}{K + P e^{10.r} - P} - 2.982\right)^2 + \left(\frac{K P e^{15.r}}{K + P e^{15.r} - P} - 3.335\right)^2 + \left(\frac{K P e^{20.r}}{K + P e^{20.r} - P} - 3.692\right)^2 + \left(\frac{K P e^{25.r}}{K + P e^{25.r} - P} - 4.068\right)^2 + \left(\frac{K P e^{30.r}}{K + P e^{30.r} - P} - 4.435\right)^2 + \left(\frac{K P e^{35.r}}{K + P e^{35.r} - P} - 4.831\right)^2 + \left(\frac{K P e^{40.r}}{K + P e^{40.r} - P} - 5.263\right)^2 + \left(\frac{K P e^{45.r}}{K + P e^{45.r} - P} - 5.674\right)^2 + \left(\frac{K P e^{50.r}}{K + P e^{50.r} - P} - 6.07\right)^2 + \left(\frac{K P e^{55.r}}{K + P e^{55.r} - P} - 6.454\right)^2 + \left(\frac{1. K P}{K + 0.} - 2.519\right)^2$$

In[13]:=
```
lsmin = FindMinimum[ls[r, K, P], {{r, 0.2}, {K, 9.0}, {P, 2.519}}, Method -> "Newton"]
```

Out[13]=

$\{0.00990818, \{r \to 0.027062, K \to 12.3612, P \to 2.4636\}\}$

The Graph

The *Least Squares Method* yields the following estimates: r = .0271 and K = 12.361.
This means that our models predicts the world population ceiling to be at approximately 12.4 billion.

In[14]:=
```
p[t_] = p[t] /. sol[[1]] /. lsmin[[2]] /. P -> 2.519
```

Out[14]=

$$\frac{30.4532\, e^{0.027062\, t}}{9.89765 + 2.4636\, e^{0.027062\, t}}$$

In[15]:=
```
p1 = ListPlot[pop, PlotStyle → {Red, AbsolutePointSize[7]}];
```

In[16]:=
```
p2 = Plot[p[t - 1950], {t, 1900, 2100}, PlotStyle → Thick,
   BaseStyle → {Bold, 14}, PlotRange → {0, 13}, AxesOrigin → {1900, 0},
   Epilog → {Dashed, Blue, Line[{{1900, lsmin[[2, 2, 2]]}, {2100, lsmin[[2, 2, 2]]}}]},
   PlotLabel → "Projected World Population 1900-2100",
   AxesLabel → {"Year", "Population (10^9)"}, GridLines → Automatic, ImageSize → 500];
```

In[17]:=
```
p2a = Plot[p[t - 1950], {t, 1950, 2005},
   PlotStyle → Thick, BaseStyle → {Bold, 14}, PlotRange → {0, 7},
   AxesOrigin → {1950, 0}, PlotLabel → "World Population 1950-2005",
   AxesLabel → {"Year", "Population (10^9)"}, GridLines → Automatic, ImageSize → 500];
```

In[18]:=

```
Show[p2a, p1]
```

Out[18]=



The inflection point of the model occurs in 2001:

In[19]:=

```
Plot[p''[t - 1950], {t, 1950, 2050}, PlotLabel → "p''(t)",
 AxesOrigin → {1950, 0}, BaseStyle → {Bold, 12}, GridLines → Automatic]
```

Out[19]=



The same function plotted up to the year 2100:

In[20]:=
```
Show[p2, p1]
```

Out[20]=



Here are the current United Nations projections:

Source: http://en.wikipedia.org/wiki/File:World-Population-1800-2100.png, retrieved 12/30/09.


Millions of people
15000
14000
13000
12000
11000
10000
9000
8000
7000
6000
5000
4000
3000
2000
1000
0
1800 1820 1840 1860 1880 1900 1920 1940 1960 1980 2000 2020 2040 2060 2080 2100
Estimated
UN High
UN Medium
UN Low
Actual